⊠연구논문

SAS/STAT를 이용하여 비선형 방정식계의 해를 구하는 방법+

남윤석
한국전자통신연구원
조태경 · 심규박
동국대학교 정보통계학과

Using SAS/STAT to Solve a System of Nonlinear Equations

Yun-Seok Nam
Electronics and Telecommunications Research Institute
Tae-Kyoung Cho · Kyu-Bark Shim
Detp. of Statistics and Information Science at Dongguk University

Abstract

There exist many computer algorithms to solve a system of nonlinear equations. But in case nonlinear equations are complex it is not easy to solve a system of nonlinear equations. In this paper we consider the method of using NLIN procedure in SAS/STAT to solve a system of nonlinear equations.

+ 본 연구는 동국대학교 전문학술지 논문 게재연구비 지원으로 이루어졌음.

1. 서론

일반적으로 다변수 함수의 Newton 방법, 고정점법, 이분법 등을 이용하여 비선형 방정식계의 근사해를 구할 수 있다. 그러나 주어진 비선형 방정식들이 매우 복잡한 경우 이러한 방법들을 사용하여 근사해를 구한다는 것이 쉽지 않다.

위에서 열거한 수치해석적인 방법 이외에도 수치계산 소프트웨어인 Mathematica 및 통계 소프트웨어인 SAS/ETS 등 수치계산 소프트웨어를 이용하여 비선형 방정식계의 근사해를 구할 수도 있으나, 대부분의 SAS 사용자들 조차도 여러 가지 이유로 Mathematica 및 SAS/ETS 등 비선형 방정식계의 근사해를 구하기 위한 소프트웨어를 소유하지 않은 경우가 많아 이를 구하는데 어려움을 겪고 있는 실정이다.

일반적으로 SAS 이용자들이 가장 많이 사용하고 있는 SAS/STAT에서도 NLIN procedure를 이용하여 비선형 회귀모형을 추정하고는 있으나, 이는 주어진 자료를 이용하여 비선형 회귀모형의 회귀모수를 추정하기 위한 절차이지 비선형 방정식의 해를 구하는 것은 아니다. 또한, SAS/STAT의 NLIN procedure에서도 비선형 방정식이 하나만 존재하는 경우 비선형 회귀모형의 모수를 추정하는 절차에 따라 근사해를 구하는 방법은 SAS manual에 소개된 바 있으나, 비선형 방정식계의 근사해를 구하기 위해 이 절차를 사용한 사례는 아직 제안된 바 없다.

본 논문에서는 SAS/STAT에서 비선형 회귀모형의 모수를 추정하는 원리를 이용하여 비선형 방정식계의 근사해를 구하는 방법을 소개하고자 한다.

이를 위해, 비선형 방정식계의 근사해를 구하기 위한 SAS/ETS 프로그램을 소개하고, SAS/STAT의 프로그램을 제안하였다. 끝으로, SAS/STAT를 이용한 방법의 우수성을 입증하기 위해 두 가지 수치 예를 들고 각 방법으로 구한 근사해와 비교하였다.

특히 비선형 방정식계에 누적확률분포함수가 포함된 경우 근사해를 구하는 것은 매우 어려우나 SAS/STAT에 내장된 확률분포함수를 이용하면 일반적인 방법보다 쉽게 근사해를 구할 수 있음도 보였다.

2. SAS를 이용한 해

n개의 미지수 x_1, x_2, $\cdots$, x_n에 관한 n개의 비선형 함수 f_1, f_2, $\cdots$, f_n으로 되어 있는 $(n \times n)$형 비선형 방정식계를 다음과 같이 나타내자.

$$\begin{aligned} f_1(x_1,\ x_2,\ \cdots,\ x_n) &= c_1 \\ f_2(x_1,\ x_2,\ \cdots,\ x_n) &= c_2 \\ &\vdots \\ f_n(x_1,\ x_2,\ \cdots,\ x_n) &= c_n \end{aligned} \tag{2-1}$$

여기서 c_1, c_2, …, c_n은 상수이다.

위의 비선형 방정식계의 근사해를 구한다는 것은 각 방정식을 동시에 만족하는 미지수 x_1, x_2, …, x_n의 근사값을 구하는 것이다.

2.1 SAS/ETS의 이용

SAS/ETS는 경제분석, 예측, 경제 및 회계모형 등 시계열 자료 분석에 주로 쓰이는 통계소프트웨어이다. 따라서, 여러 가지 변수들 사이의 연립관계를 분석하는데 유용하다. SAS/ETS에서 사용하는 procedure는 여러 가지가 있으나, MODEL procedure를 사용해 비선형 방정식계의 근사해를 구할 수 있다.
식 (2-1)과 같은 형태의 비선형 방정식계의 해를 구하기 위한 SAS/ETS 프로그램은 <그림 2-1>과 같다.

```
DATA nlineq;
INPUT x1 x2 ··· xn;
CARDS;
starting values
;
PROC MODEL DATA=nlineq;
EQ.eq1= f1(x1, x2, ···, xn) - c1;
EQ.eq2= f2(x1, x2, ···, xn) - c2;
    ⋮
EQ.eqn= fn(x1, x2, ···, xn) - cn;
SOLVE x1 x2 ··· xn/ SOLVEPRINT;
RUN;
```

< 그림 2-1 > ($n \times n$)형 비선형 방정식계의 근사해를 구하기 위한 SAS/ETS 프로그램

<그림 2-1>의 프로그램에서는 DATA set에서 변수 x_1, x_2, …, x_n에 대한 임의의 초기값(starting values)을 설정할 수 있다. 이 프로그램을 실행하여 얻어진 결과에서는 구하고자 하는 비선형 연립방정식의 근사해 이외에도 설정한 초기값에 따라 얻어진 근사값과 설정한 초기값 사이의 잔차값(residual value)이 출력된다.
프로그램에서 'EQ. 임의의 변수'를 사용하는 것은 방정식의 오차를 나타내기 위한 것이다. SOLVE문을 사용하여 근사해를 구하고자 하는 미지수들을 지정할 수 있으며,

미지수의 근사해를 출력하기 위해서 SOLVE문의 옵션으로 SOLVEPRINT를 사용한다. SOLVE문은 근사해를 구하기 위해 JACOBI, SEIDEL, 그리고 NEWTON과 같은 3가지 반복계산 방법들 중 한 가지를 선택해서 사용할 수 있다(SAS/ETS, 1994). 반복계산 방법을 지정하지 않으면 자동으로 NEWTON의 방법을 선택한다.

2.2 SAS/STAT의 이용

NLIN procedure를 이용하여 비선형 방정식계의 근사해를 구하기 위해서는 비선형 방정식계를 비선형 회귀모형의 형태로 변형시켜야 하며 지수변수(indicator variable)를 이용하면 이것이 가능하다.

식 (2-1)의 $(n \times n)$형 비선형 방정식계를 비선형 회귀모형의 형태로 만들기 위하여 미지수 x_1, x_2, $\cdots$, x_n을 추정해야할 모수라 하고, c_i를 종속변수 y_i의 관측값, 그리고 지수변수 z_i, i=1, 2, $\cdots$, n,를 독립변수라 하면 <표 2-1>과 같은 자료구조를 만들 수 있다.

< 표 2-1 > 자료구조

y	z_1	z_2	$\cdots$	z_n
c_1	1	0	$\cdots$	0
c_2	0	1	$\cdots$	0
$\vdots$	$\vdots$	$\vdots$		$\vdots$
c_n	0	0	$\cdots$	1

<표 2-1>의 자료구조를 이용하여 식 (2-1)의 ($n \times n$)형 비선형 방정식계를 다음과 같은 비선형 회귀모형의 형태로 변형시킬 수 있다.

$$y_i = \sum_{j=1}^{n} f_j(x_1,\ x_2,\ \cdots,\ x_n) z_j + \varepsilon_i,\ \ i=1,\ 2,\ \cdots, n \tag{2-2}$$

여기서 ε_i는 오차항.

식 (2-2)에 대해 NLIN procedure는 오차 제곱의 합

$$\sum_{i=1}^{n} \varepsilon_i^2 = \sum_{i=1}^{n}\left(y_i - \sum_{j=1}^{n} f_j(x_1, x_2, \cdots, x_n) z_j \right)^2 \tag{2-3}$$

을 최소로 하는 미지수 x_1, x_2, $\cdots$, x_n의 값들을 추정치로 택한다. 식 (2-3)을 최소로 하는 미지수 x_1, x_2, $\cdots$, x_n의 값들을 구하기 위해서는 $f_j(x_1, x_2, \cdots, x_n)$이 비

선형 방정식인 경우 일반적으로 반복계산을 필요로 한다. NLIN procedure는 다음의 4가지 반복계산 방법중 한 가지를 선택하여 사용할 수 있다.

· modified Gauss-Newton method (METHOD=GAUSS).
· Marquardt method (METHOD=MARQUARDT).
· gradient or steepest-descent method (METHOD=GRADIENT).
· multivariate secant or false position (METHOD=DUD).

각 방법의 자세한 내용은 Kennedy and Gentle(1980), Ralston and Jennrich(1979) 그리고 SAS/STAT(1994)등을 참조할 수 있다.

DUD(Doesn't Use Derivatives)방법을 제외한 나머지 방법들은 비선형 회귀모형에 포함된 모수들에 대한 편도함수들을 필요로 하지만 DUD방법은 편도함수를 필요로 하지 않으므로 주어진 비선형 회귀모형이 매우 복잡하여 모수들에 대한 편도함수를 구하기 어려운 경우 DUD방법을 사용하는 것이 편리하다.

SAS 프로그램에서 특정한 반복계산방법을 선택하지 않으면 NLIN procedure는 자동으로 DUD방법을 선택한다. 반복 계산에 필요한 초기값들은 PARMS문에 지정해줄 수 있다.

NLIN procedure을 이용하여 식 (2-1)의 ($n \times n$)형 비선형 방정식계에 포함된 미지수 x_1, x_2, …, x_n의 근사값을 추정하기 위한 SAS/STAT 프로그램은 <그림 2-2> 와 같다.

```
DATA nlineq;
INPUT z1 z2 ··· zn y;
CARDS ;
1 0 0  ···  0 c1
0 1 0  ···  0 c2
         ⋮
0 0 0  ···  1 cn
;
PROC NLIN DATA=nlineq;
PARMS x1=a1 x2=a2 ··· xn=an;
MODEL y= f1(x1, x2,···, xn)*z1+ f2(x1, x2,···, xn)*z2 + ···
          + fn(x1, x2,···, xn)*zn;
RUN;
```

< 그림 2-2 > ($n \times n$)형 비선형 연립방정식계의 근사해를 구하기 위한 SAS/STAT 프로그램

3. 예제

3.1 누적확률분포함수가 포함되지 않는 경우

본 논문에서 제안한 SAS/STAT에서 NLIN procedure 사용의 방법을 알아보기 위해 Faires 등(1993)이 사용한 다음의 (3×3)형 비선형 방정식계의 근사해를 구하여 보자.

$$
\begin{aligned}
&3x_1-\cos(x_1x_2)=0.5\\
&x_1^2-81(x_2+0.1)^2+\sin(x_3)=-1.06\\
&e^{(x_1x_2)}+20x_3+(10\pi-3)/3=0
\end{aligned}
\tag{3-1}
$$

Fairs 등은 비선형 방정식계 (3-1)의 근사해를 고정점법과 Newton법을 이용하여 구한 바 있다. 본 논문에서는 Fairs 등이 구한 근사해와 SAS/ETS 및 SAS/STAT를 사용하여 구한 근사해를 비교하여 보고자 한다.

비선형 방정식계 (3-1)의 근사해를 구하기 위해 $x_1=0.1$, $x_2=0.1$, $x_3=-0.1$를 초기값으로하여 작성한 SAS/ETS 프로그램 및 SAS/STAT 프로그램은 각각 <그림 3-1> 그리고 <그림 3-2>와 같다.

```
DATA nlineq;
INPUT x1 x2 x3;
CARDS;
0.1 0.1 -0.1
;
PROC MODEL DATA=nlineq;
EQ.eq1=3*x1-COS(x1*x2)-0.5;
EQ.eq2=x1**2-81*(x2+0.1)**2+SIN(x3)+1.06;
EQ.eq3=EXP(x1*x2)+20*x3+(10*(GAMMA(0.5))**2-3)/3;
SOLVE x1 x2 x3 / NEWTON SOLVEPRINT;
RUN;
```

< 그림 3-1 > 비선형 방정식계 (3-1)의 근사해를 구하기 위한 SAS/ETS 프로그램

```
DATA nlineq;
INPUT z1 z2 z3 y;
CARDS;
1 0 0  0.5
0 1 0 -1.06
0 0 1  0
;
PROC NLIN DATA=nlineq;
PARMS x1=0.1 x2=0.1 x3=-0.1;
MODEL y=(3*x1-COS(x2*x3))*z1
        +(x1**2-81*(x2+0.1)**2+SIN(x3))*z2
        +(EXP(-x1*x2)+20*x3+(10*(GAMMA(0.5))**2-3)/3)*z3;
RUN;
```

< 그림 3-2 > 비선형 방정식계 (3-1)의 근사해를 구하기 위한 SAS/STAT 프로그램

<그림 3-1>의 SAS/ETS 프로그램을 실행하면 근사해 x1=0.5, x2=-6.71E-18, x3=-0.5236 이외에도 잔차값 x1=-0.4, x2=0.1, x3=0.4236을 얻을 수 있는데, 이 값은 구하고자 하는 비선형 연립방정식의 근사해와 초기치 사이의 잔차값이다.

위의 SAS/ETS와 SAS/STAT의 프로그램을 이용하여 얻은 근사해와 고정점법과 Newton의 방법으로 구한 근사해(Faires등(1993))는 <표 3-1>과 같다.

< 표 3-1 > 비선형 방정식계 (3-1)의 근사해 비교

	반복횟수	x_1	x_2	x_3
고정점법	5	0.50000000	0.0000002	-0.52359877
Newton의 방법	5	0.50000000	0.0000000	-0.52359877
SAS/ETS이용 방법	5	0.5000	-6.71E-18	-0.5236
SAS/STAT이용 방법	5	0.5000000000	-0.0000000000	-0.5235987756

3.2 누적확률분포함수가 포함된 경우

비선형 방정식계에 누적확률분포가 포함되고 근사해가 존재하는 경우 SAS/STAT에 포함된 누적확률분포함수를 이용하면 기존의 방법보다 쉽게 근사해를 구할 수 있다.

다음과 같이 누적확률분포가 포함된 (2×2)형 비선형 방정식계의 근사해를 구하여

보자.

$$\int_{-\infty}^{x_1} \frac{1}{\sqrt{2\pi}} e^{\frac{-t^2}{2}} dt - x_2 = 0$$
$$\int_0^{x_1} \frac{t e^{-t}}{\Gamma(2)} dt - \cos(x_2) = 0 \qquad (3\text{-}2)$$

식 (3-2)의 근사해를 구하기 위한 SAS/ETS 프로그램은 <그림 3-3>과 같다.

```
DATA nlineq;
INPUT x1 x2;
CARDS;
0 0
;
PROC MODEL DATA=nlineq;
EQ.eq1=PROBNORM(x1)-x2;
EQ.eq2=PROBGAM(x1, 2)-COS(x2);
SOLVE x1 x2 / SOLVEPRINT;
RUN:
```

< 그림 3-3 > 주어진 비선형 방정식계 (3-2)의 근사해를 구하는 SAS/ETS 프로그램

<그림 3-3>을 수행한 결과 아래 <그림 3-4>와 같은 error message가 SAS의 LOG 화면에 나타났으며 SOLVE문에서 사용할 수 있는 3가지 반복계산법, JACOBI, SEIDEL, 그리고 NEWTON 모두 근사해를 구하지 못했다.

```
ERROR: The NEWTON method Jacobian matrix
of partial derivatives of the equations with
respect to the variables to be solved is singular
at observation 1, for iteration 2. The system of
equations cannot be solved.
ERROR: NEWTON method failed for observation
1 at iteration number 2. Solution aborted.
```

< 그림 3-4 > SAS/ETS를 이용하여 <그림 3-3>의 프로그램을 실시하여 얻은 결과

위에 주어진 비선형 방정식계 (3-2)의 근사해를 초기값 $x_1=0$, $x_2=0$ 으로 하여 구하기 위한 SAS/STAT 프로그램은 <그림 3-5>와 같다.

```
DATA nlineq;
INPUT z1 z2 y;
CARDS;
1 0 0
0 1 0
;
PROC NLIN DATA=nlineq;
PARMS x1=0 x2=0;
MODEL y=(PROBNORM(x1)-x2)*z1
          +(PROBGAM(x1, 2)-cos(x2))*z2;
RUN;
```

< 그림 3-5 > 주어진 비선형 방정식계 (3.2)의 근사해를 구하는 SAS/STAT 프로그램

<그림 3-5>의 프로그램을 실행한 결과 6번 반복계산을 통해서 다음과 같은 근사해를 얻었다.

$$x_1=1.894069835,\quad x_2=0.970892130$$

따라서, SAS/STAT를 사용하면 SAS/ETS에서 얻지 못한 결과를 얻을 수도 있어 이들을 서로 보완하여 사용하면 오차가 적은 근사해를 얻을 수 있음을 알았다.

4.결론

본 논문에서는 복잡한 형태의 비선형 방정식계의 근사해를 SAS/STAT를 이용하여 구하는 방법을 알아보았다. 앞에서 말한 바와 같이 비선형 방정식계의 근사해는 여러 가지 수치해석의 원리를 이용하여 구할 수도 있고, Mathematica나 SAS/ETS등 여러 수치계산 소프트웨어를 사용하여 구할 수도 있다. 그러나, Newton방법 등과 같은 수치해석적 방법은 알고리즘 구축의 어려움과 프로그램 작성의 어려움이 있고, Mathematica나 SAS/ETS등 이용하는 것은 해당 소프트웨어의 대중적 보급에 문제가 있어

사용하기에 불편함이 있다.

그러한 점에서, SAS 사용자 누구나 쉽게 접할 수 있는 SAS/STAT를 사용하여 비선형 방정식계의 근사해를 구할 수 있다는 것은 의의가 있다 하겠다.

그러나 본 논문에서 제안한 SAS/STAT의 사용으로 얻은 비선형 방정식계의 근사해에 대한 수치오차의 수리적 전개와 그 평가에 대한 문제는 다음의 연구과제로 남기고자 한다.

참고문헌

[1] Faires, J.D. and Burden, R.L.(1993), *Numerical Methods*, PWS.

[2] Kennedy, W.J. and Gentle J.E.(1980), *Statistical Computing*, New York: Marcel Dekker.

[3] Ralston, M.L. and Jennrich, R.I.(1979), "DUD, A Derivative-Free Algorithm for Nonlinear Least Squarw," *Technometrics*, 1, pp. 7-14.

[4] SAS/ETS(1994). *User's Guide*, version 6, 2nd edition, Cary, N.C., Sas Institute Inc.

[5] SAS/STAT(1994). *User's Guide*, version 6, 4th edition, Cary, N.C., Sas Institute Inc.